The Les Houches Accord

Les Houches accord May 2001 $\Rightarrow$ E Boos et al., hep-ph/0109068
or PYTHIA manual section 9.9

The LHA introduces two steps in a run, where a user can intervene:

1) at initialization, e.g. from inside `PYINIT`, do a
`CALL UPINIT`
where the user will define the character of a run by setting info in
`COMMON/HEPRUP/`

2) for each new event, e.g. from inside `PYEVNT`, do a
`CALL UPEVNT`
where the user will define the next event by setting info in
`COMMON/HEPEUP/`

PYTHIA facility enabled by `CALL PYINIT('USER',' ',' ',0D0)`

The Context



Some Specialized Generators:

- AcerMC: $\mathrm{t\bar{t}b\bar{b}}$, ...
- ALPGEN: $\mathrm{W/Z+} \leq 6\mathrm{j}$, $n\mathrm{W} + m\mathrm{Z} + k\mathrm{H+} \leq 3\mathrm{j}$, ...
- AMEGIC++: generic LO
- CompHEP: generic LO
- GRACE+Bases/Spring: generic LO+ some NLO loops
- GR@PPA: $\mathrm{b\bar{b}b\bar{b}}$
- MadCUP: $\mathrm{W/Z+} \leq 3\mathrm{j}, \mathrm{t\bar{t}b\bar{b}}$
- MadGraph+HELAS: generic LO
- MCFM: NLO $\mathrm{W/Z+} \leq 2\mathrm{j}$, WZ, WH, $\mathrm{H+} \leq 1\mathrm{j}$
- O'Mega+WHIZARD: generic LO
- VECBOS: $\mathrm{W/Z+} \leq 4\mathrm{j}$

Apologies for all unlisted programs

Initialization

```
 INTEGER MAXPUP
 PARAMETER (MAXPUP=100)
 INTEGER IDBMUP,PDFGUP,PDFSUP,IDWTUP,NPRUP,LPRUP
 DOUBLE PRECISION EBMUP,XSECUP,XERRUP,XMAXUP
 COMMON/HEPRUP/IDBMUP(2),EBMUP(2),PDFGUP(2),PDFSUP(2),IDWTUP,
&NPRUP,XSECUP(MAXPUP),XERRUP(MAXPUP),XMAXUP(MAXPUP),LPRUP(MAXPUP)
```

IDBMUP: incoming beam particles (PDG codes, $\mathrm{p} = 2212$, $\overline{\mathrm{p}} = -2212$)

EBMUP: incoming beam energies (GeV)

PDFGUP, PDFSUP: PDFLIB parton distributions (not used by PYTHIA)

IDWTUP: weighting strategy

- = 1: PYTHIA mixes and unweights events, according to known $\mathrm{d}\sigma_{\mathrm{max}}$
- = 2: PYTHIA mixes and unweights events, according to known σ_{tot}
- = **3**: unit-weight events, given by user, always to be kept
- = 4: weighted events, given by user, always to be kept
- = -1, -2, -3, -4: also allow negative $\mathrm{d}\sigma$

NPRUP: number of separate user processes

XSECUP(i): σ_{tot} for each user process

XERRUP(i): error on σ_{tot} for each user process

XMAXUP(i): $\mathrm{d}\sigma_{\mathrm{max}}$ for each user process

LPRUP(i): integer identifier for each user process

Example of UPINIT routine (from PYTHIA 6.3)

```
      SUBROUTINE UPINIT

C...Precision and commonblocks (omitted here).

C...Read info from file (you must have opened file MSTP(161)).
      IF(MSTP(161).GT.0) THEN
        READ(MSTP(161),*,END=110,ERR=110) IDBMUP(1),IDBMUP(2),EBMUP(1),
     &  EBMUP(2),PDFGUP(1),PDFGUP(2),PDFSUP(1),PDFSUP(2),IDWTUP,NPRUP
        DO 100 IPR=1,NPRUP
          READ(MSTP(161),*,END=110,ERR=110) XSECUP(IPR),XERRUP(IPR),
     &    XMAXUP(IPR),LPRUP(IPR)
  100   CONTINUE
        RETURN
C...Error or prematurely reached end of file.
  110   WRITE(MSTU(11),5000)
        STOP

C...Else not implemented.
      ELSE
        WRITE(MSTU(11),5100)
        STOP
      ENDIF

C...Formats for error printout (omitted here).

      RETURN
      END
```

The event

```
 INTEGER MAXNUP
 PARAMETER (MAXNUP=500)
 INTEGER NUP,IDPRUP,IDUP,ISTUP,MOTHUP,ICOLUP
 DOUBLE PRECISION XWGTUP,SCALUP,AQEDUP,AQCDUP,PUP,VTIMUP,SPINUP
 COMMON/HEPEUP/NUP,IDPRUP,XWGTUP,SCALUP,AQEDUP,AQCDUP,
&IDUP(MAXNUP),ISTUP(MAXNUP),MOTHUP(2,MAXNUP),ICOLUP(2,MAXNUP),
&PUP(5,MAXNUP),VTIMUP(MAXNUP),SPINUP(MAXNUP)
```

`IDPRUP`: identity of current process

`XWGTUP`: event weight (meaning depends on `IDWTUP` weighting strategy)

`SCALUP`: scale Q of parton distributions etc.

`AQEDUP`: α_{em} used in event

`AQCDUP`: α_{s} used in event

`NUP`: number of particles in event

`IDUP(i)`: PDG identity code for particle `i`

`ISTUP(i)`: status code ($-1 =$ incoming parton, $1 =$ final-state parton, $2 =$ intermediate resonance with preserved m)

`MOTHUP(j,i)`: position of one or two mothers

`PUP(j,i)`: (p_x, p_y, p_z, E, m)

`VTIMUP(i)`: invariant lifetime $c\tau$

`SPINUP(i)`: spin (helicity) information

Examples of colour flows and indices

`ICOLUP(j,i)`: colour and anticolour indices

= colour line tags, in the $N_C \to \infty$ limit, starting e.g. with number 501.

Example 1: hadronic $t\bar{t}$ production



Example 2: baryon number violation



user-process BNV not (yet) implemented in PYTHIA

(but part of internal PYTHIA SUSY machinery)

Example of UPEVNT routine (from PYTHIA 6.3)

```
      SUBROUTINE UPEVNT

C...Precision and commonblocks (omitted here).

C...Read info from file (you must have opened file MSTP(162).
      IF(MSTP(162).GT.0) THEN
        READ(MSTP(162),*,END=110,ERR=110) NUP,IDPRUP,XWGTUP,SCALUP,
     &  AQEDUP,AQCDUP
        DO 100 I=1,NUP
          READ(MSTP(162),*,END=110,ERR=110) IDUP(I),ISTUP(I),
     &    MOTHUP(1,I),MOTHUP(2,I),ICOLUP(1,I),ICOLUP(2,I),
     &    (PUP(J,I),J=1,5),VTIMUP(I),SPINUP(I)
  100   CONTINUE

        RETURN
C...Special when reached end of file or other error.
  110   NUP=0

C...Else not implemented.
      ELSE
        WRITE(MSTU(11),5000)
        STOP
      ENDIF

C...Format for error printout (omitted here).

      RETURN
      END
```

Skeleton main program

```
C...Precision and commonblocks (omitted here).

C...Files for input (specific to PYTHIA UPINIT/UPEVNT routines).
      MSTP(161)=21
      OPEN(21,FILE='sample.init',STATUS='unknown')
      MSTP(162)=22
      OPEN(22,FILE='sample.evnt',STATUS='unknown')

C...Initialize with external process.
      CALL PYINIT('USER',' ',' ',0D0)

C...Event loop; generate event; check it was obtained or quit.
      DO 130 IEV=1,NEV
        CALL PYEVNT
        IF(MSTI(51).EQ.1) GOTO 140

C...List first few events: Les Houches input and PYTHIA generated.
        IF(IEV.LE.3) CALL PYLIST(7)
        IF(IEV.LE.3) CALL PYLIST(2)

C...Insert event analysis here.

C...End of event loop. Final statistics. Done.
  130 CONTINUE
  140 CALL PYSTAT(1)
      END
```

Example of PYLIST(7) output

Event listing of user process at input (simplified)

I	IST	ID	Mothers		Colours		p_x	p_y	p_z	E	m
1	-1	21	0	0	101	102	0.000	0.000	37.808	37.808	0.000
2	-1	2	0	0	102	0	0.000	0.000	-191.351	191.351	0.000
3	2	23	1	2	0	0	1.320	-27.987	-171.365	195.951	90.806
4	1	2	1	2	101	0	-1.320	27.987	17.823	33.208	0.330
5	1	1	3	0	103	0	1.584	-55.632	-62.765	83.886	0.330
6	1	-1	3	0	0	103	-0.264	27.645	-108.601	112.065	0.330

Event listing of user process at input (simplified)

I	IST	ID	Mothers		Colours		p_x	p_y	p_z	E	m
1	-1	2	0	0	101	0	0.000	0.000	265.515	265.515	0.000
2	-1	-2	0	0	0	102	0.000	0.000	-139.629	139.629	0.000
3	2	6	1	2	101	0	60.993	15.214	111.242	217.597	176.131
4	2	-6	1	2	0	102	-60.993	-15.214	14.644	187.547	176.090
5	1	5	3	0	101	0	-17.822	-35.670	79.802	89.323	4.500
6	2	24	3	0	0	0	78.815	50.884	31.440	128.274	81.638
7	1	-5	4	0	0	102	39.830	-0.176	37.926	55.182	4.500
8	2	-24	4	0	0	0	-100.824	-15.038	-23.282	132.364	81.158
9	1	2	6	0	103	0	86.769	55.002	7.840	103.032	0.000
10	1	-1	6	0	0	103	-7.954	-4.118	23.600	25.242	0.000
11	1	-2	8	0	0	104	6.270	10.530	-18.457	22.155	0.000
12	1	1	8	0	104	0	-107.093	-25.568	-4.824	110.209	0.000

Backup: PDG Particle Codes

A. Fundamental objects

1	d	11	e^-	21	g					add $-$ sign for
2	u	12	ν_e	22	γ	32	Z'^0			antiparticle,
3	s	13	μ^-	23	Z^0	33	Z''^0			where appropriate
4	c	14	ν_μ	24	W^+	34	W'^+			
5	b	15	τ^-	25	h^0	35	H^0	37	H^+	+ diquarks, SUSY,
6	t	16	ν_τ			36	A^0	39	Graviton	technicolor, ...

B. Mesons

$100\,|q_1| + 10\,|q_2| + (2s+1)$ with $|q_1| \geq |q_2|$

particle if heaviest quark u, $\bar{s}$, c, $\bar{b}$; else antiparticle

111	π^0	311	K^0	130	K^0_L	221	η^0	411	D^+	431	D^+_s
211	π^+	321	K^+	310	K^0_S	331	η'^0	421	D^0	443	J/ψ

C. Baryons

$1000\,q_1 + 100\,q_2 + 10\,q_3 + (2s+1)$

with $q_1 \geq q_2 \geq q_3$, or Λ-like $q_1 \geq q_3 \geq q_2$

2112	n	3122	Λ^0	2224	Δ^{++}	3214	Σ^{*0}
2212	p	3212	Σ^0	1114	Δ^-	3334	Ω^-

Backup: The HEPEVT Event Record

Standard output of the *final* event.

```
 PARAMETER (NMXHEP=4000)
 COMMON/HEPEVT/NEVHEP,NHEP,ISTHEP(NMXHEP),IDHEP(NMXHEP),
&JMOHEP(2,NMXHEP),JDAHEP(2,NMXHEP),PHEP(5,NMXHEP),
&VHEP(4,NMXHEP)
 DOUBLE PRECISION PHEP, VHEP
```

`NMXHEP` = maximum number of entries
`NEVHEP` = event number

`NHEP` = number of entries in current event

`ISTHEP` = status code of entry (0 = null entry, 1 = existing entry,
2 = fragmented/decayed entry, 3 = documentation entry)
`IDHEP` = PDG particle identity (+ some internal, e.g. 92 = string)

`JMOHEP` = mother position(s)
`JDAHEP` = first and last daughter position

`PHEP` = momentum (p_x, p_y, p_z, E, m) in GeV
`VHEP` = production vertex (x, y, z, t) in mm